
Quickstart
    

Mix.install(
  [
    {:instructor, path: Path.expand("../", __DIR__)}
  ],
  config: [
    instructor: [
      adapter: Instructor.Adapters.OpenAI,
      openai: [api_key: System.fetch_env!("LB_OPENAI_API_KEY")]
    ]
  ]
)

  
    
  
  Introduction


Instructor is a library to do structured prompting with OpenAI and open source LLMs. While the idea is pretty simple, through this and the other examples you'll realize how powerful a concept this is.
So first off, what is structure prompting?
What if the LLM returned data conforming to a complicated nested schema that your code knows how to work with? Well, that's structure prompting. It's a way of cohercing the LLM to producing it's response in a known format that your downstream code can handle. In the case of Instructor, we use Ecto to provide those schemas. Good old Ecto, something you're already familiar with.
So, without further ado, let's take define a schema and take it for a spin!
defmodule Politician do
  use Ecto.Schema
  use Instructor.Validator

  @doc """
  A description of United States Politicians and the offices that they held,

  ## Fields:
  - first_name: Their first name
  - last_name: Their last name
  - offices_held:
    - office: The branch and position in government they served in
    - from_date: When they entered office or null
    - until_date: The date they left office or null
  """
  @primary_key false
  embedded_schema do
    field(:first_name, :string)
    field(:last_name, :string)

    embeds_many :offices_held, Office, primary_key: false do
      field(:office, Ecto.Enum,
        values: [:president, :vice_president, :governor, :congress, :senate]
      )

      field(:from_date, :date)
      field(:to_date, :date)
    end
  end
end
{:module, Politician, <<70, 79, 82, 49, 0, 0, 17, ...>>,
 [__schema__: 1, __schema__: 1, __schema__: 1, __schema__: 1, __schema__: 2, __schema__: 2, ...]}
Great, we have our schema describing politicans and the offices they held. Let's notice a few things that may stand out from regular Ecto usage. First, since there is no database backing the schema, it doesn't make sense to give it a primary_key. This also makes sense because there is no sensible value for the LLM to respond with.
Also we use a @doc on the schema. This isn't just for documentation purposes of the tutorial. Instructor will take any @doc tag and provide it to the LLM. Generally you'll want to use this to provide semantic descriptions of the fields and general context to the LLM to ensure you get the outputs you want. In our case we want to push the LLM to understand that we are only considering American politicians.
So, let's try asking the LLM to give us some politicians.
Instructor.chat_completion(
  model: "gpt-3.5-turbo",
  response_model: Politician,
  messages: [
    %{
      role: "user",
      content:
        "Who won the American 2020 election and what offices have they held over their career?"
    }
  ]
)
{:ok,
 %Politician{
   first_name: "Joe",
   last_name: "Biden",
   offices_held: [
     %Politician.Office{office: :president, from_date: ~D[2021-01-20], to_date: nil},
     %Politician.Office{office: :vice_president, from_date: ~D[2009-01-20], to_date: ~D[2017-01-20]},
     %Politician.Office{office: :senate, from_date: ~D[1973-01-03], to_date: ~D[2009-01-15]}
   ]
 }}
Amazing, right? Using nothing more than one of the top libraries in Elixir, Ecto, we were able to get structured output from our LLM. The data returned is ready to be processed by our regular Elixir code. Instructor supports all field types that you can express in Ecto, including embedded and associated schemas.
It's almost as if the LLM inputted the data into a Phoenix Form. All the utilities that you use to process that kind of data, you can use to process the outputs of Instructor.
One of the superpowers of this is that since we're just using changesets under the hood, you can use the same validations that you would use elsewhere in your app. Let's look at that in the next section.

  
    
  
  Validations


Instructor provides a lightweight behavior where you can define a callback function that we will call to validate the data returned by the LLM using Ecto changesets. There is nothing fancy to this API. It's just a changeset in and a changeset out.
defmodule NumberSeries do
  use Ecto.Schema
  use Instructor.Validator

  @primary_key false
  embedded_schema do
    field(:series, {:array, :integer})
  end

  @impl true
  def validate_changeset(changeset) do
    changeset
    |> Ecto.Changeset.validate_length(:series, min: 10)
    |> Ecto.Changeset.validate_change(:series, fn
      field, values ->
        if Enum.sum(values) |> rem(2) == 0 do
          []
        else
          [{field, "The sum of the series must be even"}]
        end
    end)
  end
end
{:module, NumberSeries, <<70, 79, 82, 49, 0, 0, 18, ...>>, {:validate_changeset, 1}}
In this albeit contrived example, we're going to get the LLM to return a series of numbers and validate whether it has at least 10 numbers and that the sum of the series is even.
When we ask for fewer than ten numbers, Instructor will return an error tuple with a change set that is invalid.
{:error, changeset} =
  Instructor.chat_completion(
    model: "gpt-3.5-turbo",
    response_model: NumberSeries,
    messages: [
      %{role: "user", content: "Give me the first 5 integers"}
    ]
  )

# Render our the errors down to strings.
errors =
  Ecto.Changeset.traverse_errors(changeset, fn {msg, opts} ->
    Regex.replace(~r"%{(\w+)}", msg, fn _, key ->
      opts |> Keyword.get(String.to_existing_atom(key), key) |> to_string()
    end)
  end)

{changeset.changes, errors}
{%{series: [1, 2, 3, 4, 5]},
 %{series: ["The sum of the series must be even", "should have at least 10 item(s)"]}}
Now the beauty of this is that since we have human readable errors from our validations, we can just turn around and pass those back into the LLM to get it to fix its own errors.
Instructor provides a convenience parameter, max_retries for you in the initial call which will retry against the validations up to n times.
Instructor.chat_completion(
  model: "gpt-3.5-turbo",
  response_model: NumberSeries,
  max_retries: 10,
  messages: [
    %{role: "user", content: "Give some random integers"}
  ]
)

11:15:25.188 [debug] Retrying LLM call for NumberSeries:

 "series - The sum of the series must be even\nseries - should have at least 10 item(s)"

11:15:26.662 [debug] Retrying LLM call for NumberSeries:

 "series - The sum of the series must be even"

{:ok, %NumberSeries{series: [2, 4, 6, 8, 10, 12, 14, 16, 18, 20]}}
Here we demonstrated using regular Lixar code to validate the outputs of an LLM, but we don't have to stop there. We can actually use the LLM to validate the outputs of the LLM.
In Instructor, we have provided a custom Ecto Changset validator called validate_with_llm. 
Under the hood it just uses an instructor itself to check whether the field matches some condition that you have defined in plain text.
defmodule QuestionAnswer do
  use Ecto.Schema
  use Instructor.Validator

  @primary_key false
  embedded_schema do
    field(:question, :string)
    field(:answer, :string)
  end

  @impl true
  def validate_changeset(changeset) do
    changeset
    |> validate_with_llm(:answer, "Do not say anything objectionable")
  end
end
{:module, QuestionAnswer, <<70, 79, 82, 49, 0, 0, 16, ...>>, {:validate_changeset, 1}}
%QuestionAnswer{}
|> Instructor.cast_all(%{
  question: "What is the meaning of life?",
  answer: "Sex, drugs, and rock'n roll"
})
|> QuestionAnswer.validate_changeset()
#Ecto.Changeset<
  action: nil,
  changes: %{question: "What is the meaning of life?", answer: "Sex, drugs, and rock'n roll"},
  errors: [answer: {"is invalid, Do not say anything objectionable", []}],
  data: #QuestionAnswer<>,
  valid?: false
>

  
    
  
  Record Streaming


Now if you've used chatGPT's web interface, you know that these LLMs can stream responses one token at a time. You can imagine that this is pretty easy to implement in code as you just reduce across the stream accumulating the value and appending to the log. But doing this when you're using OpenAI's function calls, and where the data is structured in JSON, it's not trivial to implement streaming.
Luckily we've done that work for you and we support two types of streaming in Instructor. The first is record streaming and the second is partial streaming.
First let's take a look at record streaming.
Record streaming is useful when you're asking the LLM for something that is naturally represented as an array of records. In this mode, we will instead of returning the full array, we'll return a stream that will emit each record once it's been completely streamed to the client, but before the next records tokens have arrived.
For example, let's take our presidents example from earlier, and we can instead ask for the first 5 presidents of the United States streaming each result as they come in.
presidents_stream =
  Instructor.chat_completion(
    model: "gpt-3.5-turbo",
    stream: true,
    response_model: {:array, Politician},
    messages: [
      %{role: "user", content: "Who are the first 5 presidents of the United States?"}
    ]
  )
#Stream<[
  enum: #Function<60.53678557/2 in Stream.transform/3>,
  funs: [#Function<48.53678557/1 in Stream.map/2>]
]>
As you can see, instead of returning the result, we return a stream which can be run to emit each of the presidents.
presidents_stream
|> Stream.each(fn {:ok, politician} -> IO.inspect(politician) end)
|> Stream.run()
%Politician{
  first_name: "George",
  last_name: "Washington",
  offices_held: [
    %Politician.Office{
      office: :president,
      from_date: ~D[1789-04-30],
      to_date: ~D[1797-03-04]
    }
  ]
}
%Politician{
  first_name: "John",
  last_name: "Adams",
  offices_held: [
    %Politician.Office{
      office: :president,
      from_date: ~D[1797-03-04],
      to_date: ~D[1801-03-04]
    }
  ]
}
%Politician{
  first_name: "Thomas",
  last_name: "Jefferson",
  offices_held: [
    %Politician.Office{
      office: :president,
      from_date: ~D[1801-03-04],
      to_date: ~D[1809-03-04]
    }
  ]
}
%Politician{
  first_name: "James",
  last_name: "Madison",
  offices_held: [
    %Politician.Office{
      office: :president,
      from_date: ~D[1809-03-04],
      to_date: ~D[1817-03-04]
    }
  ]
}
%Politician{
  first_name: "James",
  last_name: "Monroe",
  offices_held: [
    %Politician.Office{
      office: :president,
      from_date: ~D[1817-03-04],
      to_date: ~D[1825-03-04]
    }
  ]
}
:ok
An important thing to note here is that we're running the validations independently for each value in the array.  That's why the values in the stream are either {:ok, Ecto.Schema.t()} or {:error, Ecto.Changeset.t()}.
As a result, it's unclear how we can automatically do retries to fix validation errors. And therefore, when in streaming mode, it is the responsibility of the user to retry when validation errors occur. (We may revisit this decision in the future)

  
    
  
  Partial Streaming


The other streaming mode that we have an instructor is called partial streaming. In this mode, you can get back a stream that will emit the record multiple times with the fields updating as they arrive. This can be used with a schema or an array of schemas. Both are demonstrated below.
This is useful in UI applications where you want to show instant feedback to the user about what data is showing up when without giving just some indeterminant loading spinner.
Instructor.chat_completion(
  model: "gpt-3.5-turbo",
  stream: true,
  response_model: {:partial, Politician},
  messages: [
    %{role: "user", content: "Who is the first president of the United States?"}
  ]
)
|> Stream.each(fn
  {:partial, politician} -> IO.puts("[Partial]: #{inspect(politician)}")
  {:ok, politician} -> IO.puts("[Final]: #{inspect(politician)}")
end)
|> Stream.run()
[Partial]: %Politician{first_name: nil, last_name: nil, offices_held: []}
[Partial]: %Politician{first_name: nil, last_name: nil, offices_held: []}
[Partial]: %Politician{first_name: "George", last_name: nil, offices_held: []}
[Partial]: %Politician{first_name: "George", last_name: "Washington", offices_held: []}
[Partial]: %Politician{first_name: "George", last_name: "Washington", offices_held: []}
[Partial]: %Politician{first_name: "George", last_name: "Washington", offices_held: [%Politician.Office{office: nil, from_date: nil, to_date: nil}]}
[Partial]: %Politician{first_name: "George", last_name: "Washington", offices_held: [%Politician.Office{office: :president, from_date: nil, to_date: nil}]}
[Partial]: %Politician{first_name: "George", last_name: "Washington", offices_held: [%Politician.Office{office: :president, from_date: ~D[1789-04-30], to_date: nil}]}
[Partial]: %Politician{first_name: "George", last_name: "Washington", offices_held: [%Politician.Office{office: :president, from_date: ~D[1789-04-30], to_date: ~D[1797-03-04]}]}
[Final]: %Politician{first_name: "George", last_name: "Washington", offices_held: [%Politician.Office{office: :president, from_date: ~D[1789-04-30], to_date: ~D[1797-03-04]}]}
:ok
There is an important difference in this mode. Since your validations will be defined on the entirety of the object, it doesn't make sense to call the validate function until the entire record has been streamed in.
Therefore, we introduce a new output tuple in the stream compared to regular record streaming. The value can be {:partial, Ecto.Schema.t()}, and then on the last emit of the stream it can be {:error, Ecto.Changeset.t()}, or {:ok, Ecto.Schema.t()}
Like record streaming, however, using max_retries with this streaming mode does nothing and throws an error. (We may revisit this in the future when it's clear what such a behavior should do)

  
    
  
  Custom Ecto Types


Instructor supports all the Ecto types out of the box, but sometimes you need more. And that's why Instructor provides a behavior that you can implement on your own custom Ecto types. All you have to do is implement to_json_schema/0.
Whatever you return from this function will be put as the field type. See the JSONSchema Specification for more information on what you can put here. Typically you'll see people put description, type, and maybe format.
defmodule EctoURI do
  use Ecto.Type
  use Instructor.EctoType

  def type, do: :map

  # This is it, the rest is for implementing a regular old ecto type.
  def to_json_schema() do
    %{
      type: "string",
      description: "A valid URL"
    }
  end

  def cast(uri) when is_binary(uri) do
    {:ok, URI.parse(uri)}
  end

  def cast(%URI{} = uri), do: {:ok, uri}
  def cast(_), do: :error

  def load(data) when is_map(data) do
    data =
      for {key, val} <- data do
        {String.to_existing_atom(key), val}
      end

    {:ok, struct!(URI, data)}
  end

  def dump(%URI{} = uri), do: {:ok, Map.from_struct(uri)}
  def dump(_), do: :error
end
{:module, EctoURI, <<70, 79, 82, 49, 0, 0, 14, ...>>, {:dump, 1}}
Instructor.chat_completion(
  model: "gpt-3.5-turbo",
  response_model: %{url: EctoURI},
  messages: [
    %{role: "user", content: "Give me the URL for Google"}
  ]
)
{:ok,
 %{
   url: %URI{
     scheme: "https",
     authority: "www.google.com",
     userinfo: nil,
     host: "www.google.com",
     port: 443,
     path: nil,
     query: nil,
     fragment: nil
   }
 }}
And just like that, you can extend Instructor to get the LLM to return whatever you want.



  

  
    
    Philosophy 
  
    
    Local Instructor w/ llama.cpp - Instructor v0.0.5
    
    

    



  
  

    
Local Instructor w/ llama.cpp
    

Mix.install(
  [
    {:instructor, path: Path.expand("../../", __DIR__)},
    {:kino_shell, "~> 0.1.2"}
  ],
  config: [
    instructor: [
      adapter: Instructor.Adapters.Llamacpp,
      llamacpp: [
        chat_template: :mistral_instruct
      ]
    ]
  ]
)

  
    
  
  Setting up llama.cpp


The open source community has been hard at work trying to dethrone OpenAI. It turns out today there are hundreds of models that you can download on HuggingFace and run locally on your machine if you have the right hardware. One of the main ways to run these models locally is through the great project llama.cpp. You'd be surprised what a standard Macbook/Linux machine can actually run.
Instructor is designed in a way where you can swap out the provider of the LLM. Internally, it's just implemented with behavior. You can customize it by changing the configuration. In fact, look at the Mix.install of this live book to see how that's done.
config :instructor, adapter: Instructor.Adapters.Llamacpp
config :instructor, llamacpp: [chat_template: :mistral_instruct]
As of today, instructor doesn't actually run the LLM inside the BEAM. Instead, it calls out to a locally running web server that is part of the llama.cpp project. Luckily installation and configuration is easy.
Somewhere on your machine clone the Lama.CPP repo and just run make,
{_, 0} =
  System.cmd(
    "bash",
    [
      "-lc",
      """
      cd /Users/thomas/code/llama.cpp
      make
      """
    ],
    into: IO.stream()
  )

:ok
I llama.cpp build info: 
I UNAME_S:   Darwin
I UNAME_P:   arm
I UNAME_M:   arm64
I CFLAGS:    -I. -Icommon -D_XOPEN_SOURCE=600 -D_DARWIN_C_SOURCE -DNDEBUG -DGGML_USE_ACCELERATE -DACCELERATE_NEW_LAPACK -DACCELERATE_LAPACK_ILP64 -DGGML_USE_METAL  -std=c11   -fPIC -O3 -Wall -Wextra -Wpedantic -Wcast-qual -Wno-unused-function -Wshadow -Wstrict-prototypes -Wpointer-arith -Wmissing-prototypes -Werror=implicit-int -Werror=implicit-function-declaration -pthread -Wunreachable-code-break -Wunreachable-code-return -Wdouble-promotion 
I CXXFLAGS:  -I. -Icommon -D_XOPEN_SOURCE=600 -D_DARWIN_C_SOURCE -DNDEBUG -DGGML_USE_ACCELERATE -DACCELERATE_NEW_LAPACK -DACCELERATE_LAPACK_ILP64 -DGGML_USE_METAL  -std=c++11 -fPIC -O3 -Wall -Wextra -Wpedantic -Wcast-qual -Wno-unused-function -Wmissing-declarations -Wmissing-noreturn -pthread   -Wunreachable-code-break -Wunreachable-code-return -Wmissing-prototypes -Wextra-semi
I NVCCFLAGS:  
I LDFLAGS:   -framework Accelerate -framework Foundation -framework Metal -framework MetalKit 
I CC:        Apple clang version 15.0.0 (clang-1500.1.0.2.5)
I CXX:       Apple clang version 15.0.0 (clang-1500.1.0.2.5)

make: Nothing to be done for `default'.
:ok
Next, we need to actually download a model to run. One important thing to note is that llama.cpp only runs models in the GGUF file format. However, there is a great active open source community that is constantly porting the new models over to this format. Anytime you're looking for a model to run, just Google the name of the model then GGUF, and you'll usually get a result from some fellow named TheBloke.
On a fairly modest machine, you should be able to run a 7B model that is quantitized. For our example, we're going to run mistral-7b-instruct-v0.2.Q5_K_S.
A note on quantization: That Q5_K_S bit on the model represents the quantization of the model. Without getting into too much detail, roughly this represents how compressed the model is. The more compressed the model is, the lower the file size and the less RAM it takes to run. But there is slight loss in performance. If you're running a MacBook, I would suggest running the Q5 or the Q6 version of the models.

Download the model somewhere on your hard drive and then we can set up the local Llama server to run against it.
frame = Kino.Frame.new() |> Kino.render()

command =
  "/Users/thomas/code/llama.cpp/server -np 4 -cb -v -m ~/Downloads/mistral-7b-instruct-v0.2.Q6_K.gguf"

child_spec =
  Task.child_spec(fn ->
    KinoShell.print_to_frame(frame, "[KinoShell]: Starting - #{command}
")

    status_code =
      KinoShell.exec("/bin/bash", ["-lc", command], fn data ->
        KinoShell.print_to_frame(frame, data)
      end)

    color =
      if status_code == 0 do
        :yellow
      else
        :red
      end

    KinoShell.print_to_frame(frame, [
      color,
      "[KinoShell]: Command shutdown with #{status_code}
"
    ])
  end)

Kino.start_child(%{child_spec | restart: :temporary})
Kino.nothing()
defmodule President do
  use Ecto.Schema

  @primary_key false
  embedded_schema do
    field(:first_name, :string)
    field(:last_name, :string)
    field(:entered_office_date, :date)
  end
end

Instructor.chat_completion(
  response_model: President,
  messages: [
    %{role: "user", content: "Who was the first president of the United States?"}
  ]
)
{:ok,
 %President{first_name: "George", last_name: "Washington", entered_office_date: ~D[1789-04-30]}}
And there you have it. You're running Instructor against a locally running large language model.  At zero incremental cost to you.

Local Instructor w/ Ollama
    

Mix.install(
  [
    {:instructor, path: Path.expand("../../", __DIR__)},
    {:kino_shell, "~> 0.1.2"}
  ],
  config: [
    instructor: [
      adapter: Instructor.Adapters.OpenAI,
      openai: [
        api_key: "ollama",
        api_url: "http://localhost:11434"
      ]
    ]
  ]
)

  
    
  
  Setup Ollama


TODO
{_, 0} =
  System.cmd("bash", ["-lc", "# ollama pull nous-hermes2-mixtral:8x7b-dpo-q6_K"],
    into: IO.stream()
  )

:ok
defmodule President do
  use Ecto.Schema

  @primary_key false
  embedded_schema do
    field(:first_name, :string)
    field(:last_name, :string)
    field(:entered_office_date, :date)
  end
end

Instructor.chat_completion(
  mode: :json,
  model: "mistral:7b-instruct-q6_K",
  response_model: President,
  messages: [
    %{role: "user", content: "Who was the first president of the United States?"}
  ]
)
Instructor.chat_completion(
  model: "nous-hermes2-mixtral:8x7b-dpo-q6_K",
  mode: :json,
  stream: true,
  response_model: {:array, President},
  messages: [
    %{role: "user", content: "Who are the first three presidents"}
  ]
)
|> Stream.each(fn {:ok, x} -> IO.inspect(x) end)
|> Stream.run()

Instructor w/ Together.ai
    

Mix.install(
  [
    {:instructor, path: Path.expand("../../", __DIR__)}
  ],
  config: [
    instructor: [
      adapter: Instructor.Adapters.OpenAI,
      openai: [
        api_key: System.fetch_env!("LB_TOGETHER_API_KEY"),
        api_url: "https://api.together.xyz"
      ]
    ]
  ]
)

  
    
  
  Introduction


Together.ai is an LLM inference provider that is OpenAI compatible. They provide a cheap and easy way to run many the open source models that you've heard about in the cloud in an open AI compliant way that supports things like function calling, jason mode, and the other guarantees that make instructor work.
Using together.ai with instructor is as simple as pointing the API url of the OpenAIAdapter to together compute.
Mix.install(
  [
    {:instructor, path: Path.expand("../../", __DIR__)}
  ],
  config: [
    instructor: [
      adapter: Instructor.Adapters.OpenAI,
      openai: [
        api_key: System.fetch_env!("LB_TOGETHER_API_KEY"),
        api_url: "https://api.together.xyz"
      ]
    ]
  ]
)
You can sign up for an account by together.ai.
defmodule President do
  use Ecto.Schema

  @primary_key false
  embedded_schema do
    field(:first_name, :string)
    field(:last_name, :string)
    field(:entered_office_date, :date)
  end
end

Instructor.chat_completion(
  model: "mistralai/Mixtral-8x7B-Instruct-v0.1",
  response_model: President,
  messages: [
    %{role: "user", content: "Who was the first president of the United States?"}
  ]
)
{:ok,
 %President{first_name: "George", last_name: "Washington", entered_office_date: ~D[1789-04-30]}}
Instructor.chat_completion(
  model: "mistralai/Mixtral-8x7B-Instruct-v0.1",
  stream: true,
  mode: :json,
  response_model: {:array, President},
  messages: [
    %{role: "user", content: "Who are the first three presidents"}
  ]
)
|> Stream.each(fn {:ok, x} -> IO.inspect(x) end)
|> Stream.run()
%President{
  first_name: "George",
  last_name: "Washington",
  entered_office_date: ~D[1789-04-30]
}
%President{
  first_name: "John",
  last_name: "Adams",
  entered_office_date: ~D[1797-03-04]
}
%President{
  first_name: "Thomas",
  last_name: "Jefferson",
  entered_office_date: ~D[1801-03-04]
}
:ok

Text Classification
    

Mix.install(
  [
    {:instructor, path: Path.expand("../../", __DIR__)}
  ],
  config: [
    instructor: [
      adapter: Instructor.Adapters.OpenAI,
      openai: [api_key: System.fetch_env!("LB_OPENAI_API_KEY")]
    ]
  ]
)

  
    
  
  Motivation


Text classification is a common task in NLP and broadly applicable across software. Whether it be spam detection, or support ticket categorization, NLP is at the core. Historically, this required training custom, bespoke models that required collecting thousands of pre-labeled examples. With LLMs a lot of this knowledge is already encoded into the model. With proper instruction and guiding the output to a known set of classifications using GPT you can be up and running with a text classification model in no time.
Hell, you can even use instructor to help generate the training set to train your own more efficient model. But let's not get ahead of ourselves, there's more on that later in the tutorials.

  
    
  
  Binary Text Classification


Spam detection is a classic example of binary text classification. It's as simple as returning a true / false of whether an example is in the class. This is pretty trivial to implement in instructor.
defmodule SpamPrediction do
  use Ecto.Schema
  use Instructor.Validator

  @doc """
  ## Field Descriptions:
  - class: Whether or not the email is spam.
  - reason: A short, less than 10 word rationalization for the classification.
  - score: A confidence score between 0.0 and 1.0 for the classification.
  """
  @primary_key false
  embedded_schema do
    field(:class, Ecto.Enum, values: [:spam, :not_spam])
    field(:reason, :string)
    field(:score, :float)
  end

  @impl true
  def validate_changeset(changeset) do
    changeset
    |> Ecto.Changeset.validate_number(:score,
      greater_than_or_equal_to: 0.0,
      less_than_or_equal_to: 1.0
    )
  end
end

is_spam? = fn text ->
  Instructor.chat_completion(
    model: "gpt-3.5-turbo",
    response_model: SpamPrediction,
    max_retries: 3,
    messages: [
      %{
        role: "user",
        content: """
        Your purpose is to classify customer support emails as either spam or not.
        This is for a clothing retail business.
        They sell all types of clothing.

        Classify the following email: 
        ```
        #{text}
        ```
        """
      }
    ]
  )
end

is_spam?.("Hello I am a Nigerian prince and I would like to send you money")
{:ok, %SpamPrediction{class: :spam, reason: "Nigerian prince email", score: 0.95}}
We don't have to stop just at a boolean inclusion, we can also easily extend this idea to multiple categories or classes that we can classify the text into. In this example, let's consider classifying support emails. We want to know whether it's a general_inquiry, billing_issue, or a technical_issue perhaps it rightly fits in multiple classes. This can be useful if we want to cc' specialized support agents when intersecting customer issues occur
We can leverage Ecto.Enum to define a schema that restricts the LLM output to be a list of those values. We can also provide a @doc description to help guide the LLM with the semantic understanding of what these classifications ought to represent.
defmodule EmailClassifications do
  use Ecto.Schema

  @doc """
  A classification of a customer support email.

  technical_issue - whether the user is having trouble accessing their account.
  billing_issue - whether the customer is having trouble managing their billing or credit card
  general_inquiry - all other issues
  """
  @primary_key false
  embedded_schema do
    field(:tags, {:array, Ecto.Enum},
      values: [:general_inquiry, :billing_issue, :technical_issue]
    )
  end
end

classify_email = fn text ->
  {:ok, %{tags: result}} =
    Instructor.chat_completion(
      model: "gpt-3.5-turbo",
      response_model: EmailClassifications,
      messages: [
        %{
          role: "user",
          content: "Classify the following text: #{text}"
        }
      ]
    )

  result
end

classify_email.("My account is locked and I can't access my billing info.")
[:technical_issue, :billing_issue]

Extracting Action Items from Meeting Transcripts
    

Mix.install(
  [
    {:instructor, path: Path.expand("../../", __DIR__)},
    {:kino, "~> 0.12.0"}
  ],
  config: [
    instructor: [
      adapter: Instructor.Adapters.OpenAI,
      openai: [api_key: System.fetch_env!("LB_OPENAI_API_KEY")]
    ]
  ]
)

  
    
  
  Motivation


This example shouldn't be foreign to any of you. I'm sure each day you log in, join a Zoom call, and have a meeting about what y'all are going to do this upcoming week. It is then the work of some product manager to translate this into a JIRA board so that you can track your progress throughout the week.
The bane of most engineers existence... We can automate this.

  
    
  
  The Schema


Let's start by defining a schema for the tickets and the subtasks that might exist within them. There will be a priority, description and a set of dependencies between the tasks for the tickets.
defmodule MeetingNotes do
  use Ecto.Schema
  use Instructor.Validator

  @doc """
  Tickets correctly resolved from a meeting transcription.

  Tickets have a name, priority, useful description, and assignees.
  They may also have subtasks that share the same id spaces as the tickets.
  A ticket also may have references to Tickets and Subtasks that are a blocking dependency
  for completing the ticket.
  """
  @primary_key false
  embedded_schema do
    embeds_many :tickets, Ticket do
      field(:name, :string)
      field(:description, :string)
      field(:priority, Ecto.Enum, values: [:high, :medium, :low])
      field(:assignees, {:array, :string})

      embeds_many :subtasks, SubTasks do
        field(:name, :string)
      end

      field(:dependencies, {:array, :binary_id})
    end
  end
end
{:module, MeetingNotes, <<70, 79, 82, 49, 0, 0, 17, ...>>,
 [__schema__: 1, __schema__: 1, __schema__: 1, __schema__: 1, __schema__: 2, __schema__: 2, ...]}
One thing interesting to note here is that although we could use proptorecto associations, in this case it's rather tedious. Since our dependencies can either be on tickets or on subtasks, you would typically have to have a polymorphic relationship.
While Instructor fully supports doing that, we can instead just create embedded schemas for the subtasks and define that the subtasks and the tickets share the same ID space in our doc comment. This will steer the LLM to produce foreign key relations in a polymorphic way.
One also might choose to solve this modeling problem by denormalizing the tickets and subtasks into its own task type and get the LLM to not only have the dependencies listed but also any subtasks that a ticket might have through association instead of embedding.
Either way works, but in this example we prefer the embedded method because it produces a prettier output and we can lean on the LLM a little more.
Now, let's extract some JIRA tickets.
generate_tickets = fn transcript ->
  Instructor.chat_completion(
    model: "gpt-3.5-turbo",
    response_model: MeetingNotes,
    messages: [
      %{role: "system", content: "The following is a transcript of a meeting..."},
      %{
        role: "user",
        content: "Create the action items for the following transcript: #{transcript}"
      }
    ]
  )
end

{:ok, %{tickets: tickets}} =
  generate_tickets.("""
    Alice: Hey team, we have several critical tasks we need to tackle for the upcoming release. First, we need to work on improving the authentication system. It's a top priority.
    
    Bob: Got it, Alice. I can take the lead on the authentication improvements. Are there any specific areas you want me to focus on?
    
    Alice: Good question, Bob. We need both a front-end revamp and back-end optimization. So basically, two sub-tasks.
    
    Carol: I can help with the front-end part of the authentication system.
    
    Bob: Great, Carol. I'll handle the back-end optimization then.
    
    Alice: Perfect. Now, after the authentication system is improved, we have to integrate it with our new billing system. That's a medium priority task.
    
    Carol: Is the new billing system already in place?
    
    Alice: No, it's actually another task. So it's a dependency for the integration task. Bob, can you also handle the billing system?
    
    Bob: Sure, but I'll need to complete the back-end optimization of the authentication system first, so it's dependent on that.
    
    Alice: Understood. Lastly, we also need to update our user documentation to reflect all these changes. It's a low-priority task but still important.
    
    Carol: I can take that on once the front-end changes for the authentication system are done. So, it would be dependent on that.
    
    Alice: Sounds like a plan. Let's get these tasks modeled out and get started.
  """)
{:ok,
 %MeetingNotes{
   tickets: [
     %MeetingNotes.Ticket{
       id: "1",
       name: "Improve authentication system",
       description: "Work on improving both front-end and back-end of the authentication system",
       priority: :high,
       assignees: ["Bob"],
       subtasks: [
         %MeetingNotes.Ticket.SubTasks{id: "1.1", name: "Front-end revamp"},
         %MeetingNotes.Ticket.SubTasks{id: "1.2", name: "Back-end optimization"}
       ],
       dependencies: []
     },
     %MeetingNotes.Ticket{
       id: "2",
       name: "Integrate authentication system with billing system",
       description: "Integration of the improved authentication system with the new billing system",
       priority: :medium,
       assignees: ["Bob"],
       subtasks: [],
       dependencies: ["1", "3"]
     },
     %MeetingNotes.Ticket{
       id: "3",
       name: "Implement new billing system",
       description: "Implement the new billing system",
       priority: :medium,
       assignees: ["Bob"],
       subtasks: [],
       dependencies: ["1.2"]
     },
     %MeetingNotes.Ticket{
       id: "4",
       name: "Update user documentation",
       description: "Update the user documentation to reflect the changes",
       priority: :low,
       assignees: ["Carol"],
       subtasks: [],
       dependencies: ["1.1"]
     }
   ]
 }}
The results look good. We can use Kino to then render out all the dependencies of our tickets.
generate_tickets_diagram = fn tickets ->
  subtasks = Enum.flat_map(tickets, & &1.subtasks)
  all_tasks = subtasks ++ tickets

  ticket_nodes =
    tickets
    |> Enum.map_join("\n", fn t ->
      subtask_relations =
        t.subtasks
        |> Enum.map_join("\n", fn st ->
          """
          "#{t.name}" ||--o| "#{st.name}" : "Has Subtask"
          """
        end)

      dependency_relations =
        t.dependencies
        |> Enum.map_join("\n", fn d ->
          dt = all_tasks |> Enum.find(&(&1.id == d))

          if dt do
            """
            "#{t.name}" ||--o| "#{dt.name}" : "Depends on"
            """
          else
            ""
          end
        end)

      """
        "#{t.name}" {
          priority #{t.priority}
          assignees #{Enum.join(t.assignees, ", ")}
        }

        #{subtask_relations}
        #{dependency_relations}
      """
    end)

  subtask_nodes =
    tickets
    |> Enum.flat_map(& &1.subtasks)
    |> Enum.map_join("\n", fn st ->
      """
        "#{st.name}"
      """
    end)

  Kino.Mermaid.new("""
    erDiagram
      #{ticket_nodes}
      #{subtask_nodes}
  """)
end

generate_tickets_diagram.(tickets)
  erDiagram
      "Improve authentication system" {
    priority high
    assignees Bob
  }

  "Improve authentication system" ||--o| "Front-end revamp" : "Has Subtask"

"Improve authentication system" ||--o| "Back-end optimization" : "Has Subtask"

  

  "Integrate authentication system with billing system" {
    priority medium
    assignees Bob
  }

  
  "Integrate authentication system with billing system" ||--o| "Improve authentication system" : "Depends on"

"Integrate authentication system with billing system" ||--o| "Implement new billing system" : "Depends on"


  "Implement new billing system" {
    priority medium
    assignees Bob
  }

  
  "Implement new billing system" ||--o| "Back-end optimization" : "Depends on"


  "Update user documentation" {
    priority low
    assignees Carol
  }

  
  "Update user documentation" ||--o| "Front-end revamp" : "Depends on"


      "Front-end revamp"

  "Back-end optimization"

Converting Text into Explorer DataFrames
    

Mix.install(
  [
    {:instructor, path: Path.expand("../../", __DIR__)},
    {:explorer, "~> 0.7.2"},
    {:kino, "~> 0.12.0"},
    {:kino_explorer, "~> 0.1.13"}
  ],
  config: [
    instructor: [
      adapter: Instructor.Adapters.OpenAI,
      openai: [api_key: System.fetch_env!("LB_OPENAI_API_KEY")]
    ]
  ]
)

  
    
  
  Motivation


Sometimes we come across text that can be naturally extracted into structured data. Although we don't necessarily know how many tables, and what the schemas are represented in the data. Using Instructor we can let the LLM determine the tables and schemas for us, pulling the results into a dataframe which we can then use for further analysis.

  
    
  
  The Schema


Notice from this example we want resulting data structure to have data frames. That's not a default Ecto type so we're gonna have to do things a little different here.
Luckily, Ecto allows us to define our own custom types in which we can override how the values are casted. And Instructor provides us a behavior which tells us how to represent it in a JSONSchema so the LLM can understand it.
In our case, we're going to create a type for a table of data, which will be represented as a CSV string with respect to the LLM. Our custom type will then parse the string when it is casted by Ecto and parse the CSV into an Explorer.DataFrame.
defmodule Ecto.CSVDataFrame do
  use Ecto.Type
  use Instructor.EctoType

  def type, do: :string

  def cast(csv_str) when is_binary(csv_str) do
    df = Explorer.DataFrame.load_csv!(csv_str)
    {:ok, df}
  end

  def cast(%Explorer.DataFrame{} = df), do: {:ok, df}
  def cast(_), do: :error

  def to_json_schema(),
    do: %{
      type: "string",
      description: "A CSV representation of a data table"
    }

  def dump(x), do: {:ok, x}
  def load(x), do: {:ok, x}
end
{:module, Ecto.CSVDataFrame, <<70, 79, 82, 49, 0, 0, 12, ...>>, {:load, 1}}
Now that we have our data type, we can use it in our embedded ecto schema, just as we would have in any other example. If everything works correctly, we should get some data frames out.
defmodule Database do
  use Ecto.Schema
  use Instructor.Validator

  @doc """
  The extracted database will contain one or more tables with data as csv formatted with ',' delimiters
  """
  @primary_key false
  embedded_schema do
    embeds_many :tables, DataFrame, primary_key: false do
      field(:name, :string)
      field(:data, Ecto.CSVDataFrame)
    end
  end
end
{:module, Database, <<70, 79, 82, 49, 0, 0, 16, ...>>,
 [__schema__: 1, __schema__: 1, __schema__: 1, __schema__: 1, __schema__: 2, __schema__: 2, ...]}
Perfect. And now let's test it out on some example text.
extract_df = fn text ->
  Instructor.chat_completion(
    model: "gpt-3.5-turbo",
    response_model: Database,
    messages: [
      %{
        role: "system",
        content: """
        Map this data into one or more dataframes and correctly define the columns and data
        """
      },
      %{
        role: "user",
        content: "#{text}"
      }
    ]
  )
end

{:ok, db} =
  extract_df.("""
    My name is John and I am 25 years old. I live in
    New York and I like to play basketball. His name is
    Mike and he is 30 years old. He lives in San Francisco
    and he likes to play baseball. Sarah is 20 years old
    and she lives in Los Angeles. She likes to play tennis.
    Her name is Mary and she is 35 years old.
    She lives in Chicago.
    
    On one team 'Tigers' the captain is John and there are 12 players.
    On the other team 'Lions' the captain is Mike and there are 10 players.
  """)

Kino.Layout.tabs(
  for table <- db.tables do
    {table.name, table.data}
  end
)

GPT4-Vision - Extracting Data from Images
    

Mix.install(
  [
    {:instructor, path: Path.expand("../../", __DIR__)},
    {:kino, "~> 0.12.3"}
  ],
  config: [
    instructor: [
      adapter: Instructor.Adapters.OpenAI,
      openai: [api_key: System.fetch_env!("LB_OPENAI_API_KEY")]
    ]
  ]
)

  
    
  
  Motivation


In recent months, the latest AI researcher labs have turned LLMs into multimodal models. What this means is that no longer do they just interpret text, but they can also interpret images. One example of this provided by OpenAI is the GPT4-vision-preview model. With no extra work, you can now provide images into your prompts with instructor and still do the normal structured extractions that you're used to.
In the following example, we will extract product details from a screenshot of a Shopify store.
[image: ]
image = Kino.FS.file_path("shopify-screenshot.png") |> File.read!()
base64_image = "data:image/png;base64," <> Base.encode64(image)

defmodule Product do
  use Ecto.Schema

  @primary_key false
  embedded_schema do
    field(:name, :string)
    field(:price, :decimal)
    field(:currency, Ecto.Enum, values: [:usd, :gbp, :eur, :cny])
    field(:color, :string)
  end
end

{:ok, result} =
  Instructor.chat_completion(
    model: "gpt-4-vision-preview",
    response_model: Product,
    mode: :md_json,
    messages: [
      %{
        role: "user",
        content: [
          %{type: "text", text: "What is the product details of the following image?"},
          %{type: "image_url", image_url: %{url: base64_image, detail: "high"}}
        ]
      }
    ],
    max_tokens: 1800
  )

result
%Product{
  name: "Thomas Wooden Railway Thomas the Tank Engine",
  price: Decimal.new("33.0"),
  currency: :usd,
  color: "Blue"
}

Instructor 
    



      
 Structured prompting for LLMs. Instructor is a spiritual port of the great Instructor Python Library by @jxnlco, check out his talk on YouTube.
 The Instructor library is useful for coaxing an LLM to return JSON that maps to an Ecto schema that you provide, rather than the default unstructured text output. If you define your own validation logic, Instructor can automatically retry prompts when validation fails (returning natural language error messages to the LLM, to guide it when making corrections).
Instructor is designed to be used with the OpenAI API by default, but it also works with llama.cpp and Bumblebee (Coming Soon!) by using an extendable adapter behavior.
At its simplest, usage is pretty straightforward: 
	Create an ecto schema, with a @doc string that explains the schema definition to the LLM. 
	Define a validate_changeset/1 function on the schema, and use the Instructor.Validator macro in order for Instructor to know about it.
	Make a call to Instructor.chat_completion/1 with an instruction for the LLM to execute.

You can use the max_retries parameter to automatically, iteratively go back and forth with the LLM to try fixing validation errorswhen they occur.
defmodule SpamPrediction do
  use Ecto.Schema
  use Instructor.Validator

  @doc """
  ## Field Descriptions:
  - class: Whether or not the email is spam.
  - reason: A short, less than 10 word rationalization for the classification.
  - score: A confidence score between 0.0 and 1.0 for the classification.
  """
  @primary_key false
  embedded_schema do
    field(:class, Ecto.Enum, values: [:spam, :not_spam])
    field(:reason, :string)
    field(:score, :float)
  end

  @impl true
  def validate_changeset(changeset) do
    changeset
    |> Ecto.Changeset.validate_number(:score,
      greater_than_or_equal_to: 0.0,
      less_than_or_equal_to: 1.0
    )
  end
end

is_spam? = fn text ->
  Instructor.chat_completion(
    model: "gpt-3.5-turbo",
    response_model: SpamPrediction,
    max_retries: 3,
    messages: [
      %{
        role: "user",
        content: """
        Your purpose is to classify customer support emails as either spam or not.
        This is for a clothing retail business.
        They sell all types of clothing.

        Classify the following email: 
        ```
        #{text}
        ```
        """
      }
    ]
  )
end

is_spam?.("Hello I am a Nigerian prince and I would like to send you money")

# => {:ok, %SpamPrediction{class: :spam, reason: "Nigerian prince email scam", score: 0.98}}
Check out our Quickstart Guide for more code snippets that you can run locally (in Livebook). Or, to get a better idea of the thinking behind Instructor, read more about our Philosophy & Motivations.
Optionally, you can also customize the your llama.cpp calls (with defaults shown):
llamacpp
config :instructor, adapter: Instructor.Adapters.Llamacpp
config :instructor, :llamacpp,
    chat_template: :mistral_instruct,
    api_url: "http://localhost:8080/completion"
````

      


      
        Summary


  
    Functions
  


    
      
        cast_all(schema, params)

      


        Casts all the parameters in the params map to the types defined in the types map.
This works both with Ecto Schemas and maps of Ecto types (see Schemaless Ecto).



    


    
      
        chat_completion(params, config \\ nil)

      


        Create a new chat completion for the provided messages and parameters.



    





      


      
        Functions

        


  
    
      
      Link to this function
    
    cast_all(schema, params)


      
       
       View Source
     


  


  

Casts all the parameters in the params map to the types defined in the types map.
This works both with Ecto Schemas and maps of Ecto types (see Schemaless Ecto).

  
    
  
  Examples


When using a full Ecto Schema
iex> Instructor.cast_all(%{
...>   data: %Instructor.Demos.SpamPrediction{},
...>   types: %{
...>     class: :string,
...>     score: :float
...>   }
...> }, %{
...>   class: "spam",
...>   score: 0.999
...> })
%Ecto.Changeset{
  action: nil,
  changes: %{
    class: "spam",
    score: 0.999
  },
  errors: [],
  data: %Instructor.Demos.SpamPrediction{
    class: :spam,
    score: 0.999
  },
  valid?: true
}
When using a map of Ecto types
iex> Instructor.cast_all(%Instructor.Demo.SpamPrediction{}, %{
...>   class: "spam",
...>   score: 0.999
...> })
%Ecto.Changeset{
  action: nil,
  changes: %{
    class: "spam",
    score: 0.999
  },
  errors: [],
  data: %{
    class: :spam,
    score: 0.999
  },
  valid?: true
}
and when using raw Ecto types,
iex> Instructor.cast_all({%{},%{name: :string}, %{
...>   name: "George Washington"
...> })
%Ecto.Changeset{
  action: nil,
  changes: %{
    name: "George Washington",
  },
  errors: [],
  data: %{
    name: "George Washington",
  },
  valid?: true
}

  



    

  
    
      
      Link to this function
    
    chat_completion(params, config \\ nil)


      
       
       View Source
     


  


  

      

          @spec chat_completion(Keyword.t(), any()) ::
  {:ok, Ecto.Schema.t()}
  | {:error, Ecto.Changeset.t()}
  | {:error, String.t()}
  | Stream.t()


      


Create a new chat completion for the provided messages and parameters.
The parameters are passed directly to the LLM adapter.
By default they shadow the OpenAI API parameters.
For more information on the parameters, see the OpenAI API docs.
Additionally, the following parameters are supported:
	:response_model - The Ecto schema to validate the response against, or a valid map of Ecto types (see Schemaless Ecto).
	:stream - Whether to stream the response or not. (defaults to false)
	:validation_context - The validation context to use when validating the response. (defaults to %{})
	:mode - The mode to use when parsing the response, :tools, :json, :md_json (defaults to :tools), generally speaking you don't need to change this unless you are not using OpenAI.
	:max_retries - The maximum number of times to retry the LLM call if it fails, or does not pass validations.             (defaults to `0`)



  
    
  
  Examples


iex> Instructor.chat_completion(%{
...>   model: "gpt-3.5-turbo",
...>   response_model: Instructor.Demos.SpamPrediction,
...>   messages: [
...>     %{
...>       role: "user",
...>       content: "Classify the following text: Hello, I am a Nigerian prince and I would like to give you $1,000,000."
...>     }
...> })
{:ok,
    %Instructor.Demos.SpamPrediction{
        class: :spam
        score: 0.999
    }}
When you're using Instructor in Streaming Mode, instead of returning back a tuple, it will return back a stream that emits tuples.
There are two main streaming modes available. array streaming and partial streaming.
Partial streaming will emit the record multiple times until it's complete.
iex> Instructor.chat_completion(%{
...>   model: "gpt-3.5-turbo",
...>   response_model: {:partial, %{name: :string, birth_date: :date}}
...>   messages: [
...>     %{
...>       role: "user",
...>       content: "Who is the first president of the United States?"
...>     }
...> }) |> Enum.to_list()
[
  {:partial, %{name: "George Washington"}},
  {:partial, %{name: "George Washington", birth_date: ~D[1732-02-22]}},
  {:ok, %{name: "George Washington", birth_date: ~D[1732-02-22]}}
]
Whereas with array streaming, you can ask the LLM to return multiple instances of your Ecto schema,
and instructor will emit them one at a time as they arrive in complete form and validated.
iex> Instructor.chat_completion(%{
...>   model: "gpt-3.5-turbo",
...>   response_model: {:array, %{name: :string, birth_date: :date}}
...>   messages: [
...>     %{
...>       role: "user",
...>       content: "Who are the first 5 presidents of the United States?"
...>     }
...> }) |> Enum.to_list()

[
  {:ok, %{name: "George Washington", birth_date: ~D[1732-02-22]}},
  {:ok, %{name: "John Adams", birth_date: ~D[1735-10-30]}},
  {:ok, %{name: "Thomas Jefferson", birth_date: ~D[1743-04-13]}},
  {:ok, %{name: "James Madison", birth_date: ~D[1751-03-16]}},
  {:ok, %{name: "James Monroe", birth_date: ~D[1758-04-28]}}
]
If there's a validation error, it will return an error tuple with the change set describing the errors.
iex> Instructor.chat_completion(%{
...>   model: "gpt-3.5-turbo",
...>   response_model: Instructor.Demos.SpamPrediction,
...>   messages: [
...>     %{
...>       role: "user",
...>       content: "Classify the following text: Hello, I am a Nigerian prince and I would like to give you $1,000,000."
...>     }
...> })
{:error,
    %Ecto.Changeset{
        changes: %{
            class: "foobar",
            score: -10.999
        },
        errors: [
            class: {"is invalid", [type: :string, validation: :cast]}
        ],
        valid?: false
    }}

Instructor.Adapter behaviour
    



      
Behavior for Instructor.Adapter.

      


      
        Summary


  
    Callbacks
  


    
      
        chat_completion(list, any)

      


    





      


      
        Callbacks

        


  
    
      
      Link to this callback
    
    chat_completion(list, any)


      
       
       View Source
     


  


  

      

          @callback chat_completion([Keyword.t()], any()) :: any()


      



  


        

      



  

  
    
    Instructor.Adapters.Llamacpp - Instructor v0.0.5
    
    

    



  
  

    
Instructor.Adapters.Llamacpp 
    



      
Runs against the llama.cpp server. To be clear this calls the llamacpp specific
endpoints, not the open-ai compliant ones.
You can read more about it here:
  https://github.com/ggerganov/llama.cpp/tree/master/examples/server

      


      
        Summary


  
    Functions
  


    
      
        chat_completion(params, config \\ nil)

      


        Run a completion against the llama.cpp server, not the open-ai compliant one.
This gives you more specific control over the grammar, and the ability to
provide other parameters to the specific LLM invocation.



    





      


      
        Functions

        


    

  
    
      
      Link to this function
    
    chat_completion(params, config \\ nil)


      
       
       View Source
     


  


  

Run a completion against the llama.cpp server, not the open-ai compliant one.
This gives you more specific control over the grammar, and the ability to
provide other parameters to the specific LLM invocation.
You can read more about the parameters here:
  https://github.com/ggerganov/llama.cpp/tree/master/examples/server

  
    
  
  Examples


  iex> Instructor.chat_completion(%{
  ...>   model: "mistral-7b-instruct",
  ...>   messages: [
  ...>     %{ role: "user", content: "Classify the following text: Hello I am a Nigerian prince and I would like to send you money!" },
  ...>   ],
  ...>   response_model: response_model,
  ...>   temperature: 0.5,
  ...> })

  


        

      



  

  
    
    Instructor.Adapters.OpenAI - Instructor v0.0.5
    
    

    



  
  

    
Instructor.Adapters.OpenAI 
    



      
Documentation for Instructor.Adapters.OpenAI.

      





  

  
    
    Instructor.EctoType - Instructor v0.0.5
    
    

    



  
  

    
Instructor.EctoType behaviour
    



      
Instructor.EctoType is a behaviour that lets your implement your own custom Ecto.Type
  that works natively with Instructor.

  
    
  
  Example


  defmodule MyCustomType do
    use Ecto.Type
    use Instructor.EctoType

    # ... See `Ecto.Type` for implementation details

    def to_json_schema() do
      %{
        type: "string",
        format: "email"
      }
    end
  end

      


      
        Summary


  
    Callbacks
  


    
      
        to_json_schema()

      


    





      


      
        Callbacks

        


  
    
      
      Link to this callback
    
    to_json_schema()


      
       
       View Source
     


  


  

      

          @callback to_json_schema() :: map()


      



  


        

      



  

  
    
    Instructor.JSONSchema - Instructor v0.0.5
    
    

    



  
  

    
Instructor.JSONSchema 
    




      
        Summary


  
    Functions
  


    
      
        from_ecto_schema(ecto_schema)

      


          Generates a JSON Schema from an Ecto schema.



    





      


      
        Functions

        


  
    
      
      Link to this function
    
    from_ecto_schema(ecto_schema)


      
       
       View Source
     


  


  

  Generates a JSON Schema from an Ecto schema.
  Note: This will output a correct JSON Schema for the given Ecto schema, but
  it will not necessarily be optimal, nor support all Ecto types.

  


        

      



  

  
    
    Instructor.JSONStreamParser - Instructor v0.0.5
    
    

    



  
  

    
Instructor.JSONStreamParser 
    




      
        Summary


  
    Functions
  


    
      
        parse(chunks)

      


    





      


      
        Functions

        


  
    
      
      Link to this function
    
    parse(chunks)


      
       
       View Source
     


  


  


  


        

      



  

  
    
    Instructor.Validator - Instructor v0.0.5
    
    

    



  
  

    
Instructor.Validator behaviour
    



      
By default you'll get whatever OpenAI returns.
This behavior provides a hook for you to critique the response using standard ecto changesets validations.
This can be used in conjuction with the :max_retries parameter to Instructor.chat_completion/1 to retry the completion until it passes your validation.

  
    
  
  Examples


  defmodule Instructor.Demos.SpamPrediction do
use Ecto.Schema

@primary_key false
schema "spam_prediction" do
    field :class, :string
    field :score, :float
end

@impl true
def validate_changeset(changeset) do
    changeset
    |> validate_number(:score, less_than_or_equal_to: 1.0)
end
  end
  iex> Instructor.chat_completion(%{
  ...>   model: "gpt-3.5-turbo",
  ...>   response_model: Instructor.Demos.SpamPrediction,
  ...>   max_retries: 1,
  ...>   messages: [
  ...>     %{
  ...>       role: "user",
  ...>       content: "Classify the following text: Hello, I am a Nigerian prince and I would like to give you $1,000,000."
  ...>     }
  ...> })
  {:error, %Ecto.Changeset{
  action: nil,
  changes: %{},
  errors: [
      score: {"is invalid", [validation: :number, validation_opts: [less_than_or_equal_to: 1.0]]}
  ],
  data: %Instructor.Demos.SpamPrediction{
      class: nil,
      score: nil
  },
  valid?: false
  }}

      


      
        Summary


  
    Callbacks
  


    
      
        validate_changeset(t)

      


    


    
      
        validate_changeset(t, t)

      


    





  
    Functions
  


    
      
        validate_with_llm(changeset, field, statement, opts \\ [])

      


        Validate a changeset field using a language model



    





      


      
        Callbacks

        


  
    
      
      Link to this callback
    
    validate_changeset(t)


      
       
       View Source
     


      (optional)

  


  

      

          @callback validate_changeset(Ecto.Changeset.t()) :: Ecto.Changeset.t()


      



  



  
    
      
      Link to this callback
    
    validate_changeset(t, t)


      
       
       View Source
     


      (optional)

  


  

      

          @callback validate_changeset(Ecto.Changeset.t(), Map.t()) :: Ecto.Changeset.t()


      



  


        

      

      
        Functions

        


    

  
    
      
      Link to this function
    
    validate_with_llm(changeset, field, statement, opts \\ [])


      
       
       View Source
     


  


  

Validate a changeset field using a language model

  
    
  
  Example


  defmodule QuestionAnswer do
use Ecto.Schema

@primary_key false
embedded_schema do
    field :question, :string
    field :answer, :string
end

@impl true
def validate_changeset(changeset) do
    changeset
    |> validate_with_llm(:answer, "do not say anything objectionable")
end
  end

  


        

      



  

  
    
    Instructor.Validator.Validation - Instructor v0.0.5
    
    

    



  
  

    
Instructor.Validator.Validation 
    




      
        Summary


  
    Functions
  


    
      
        %Instructor.Validator.Validation{}

      


        Validate if an attribute is correct and if not, return an error message



    





      


      
        Functions

        


  
    
      
      Link to this function
    
    %Instructor.Validator.Validation{}


      
       
       View Source
     


      (struct)

  


  

Validate if an attribute is correct and if not, return an error message

